Pensieve header: Verifying the “Big G Lemmas”.

## Implementing $g_1$

```
ϵ /: ϵ^2 = 0;
PBWRule = {u → 1, c → 2, w → 3};
B[U@c, U@w] = -(B[U@w, U@c] = U@w);
B[U@u, U@c] = -(B[U@c, U@u] = U@u);
B[U@w, U@u] = -(B[U@u, U@w] = b U[] - 2 ϵ U@c);
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[]; UU[l_, r___] := U[l] ** UU[r];
U_i_[ℰ_] := ℰ /. {b → b_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
```

```
B[x_, x_] = 0;
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, y_] := x ** y - y ** x;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
U[x_] ** U[y_] := If[x ≺ y, U[x, y], U[y, x] + B[U@x, U@y]];
U[x_] ** U[y1_, yy__] := If[x ≤ y1, U[x, y1, yy], (U@x ** U@y1) ** U@yy];
U[xx__, xn_] ** U[yy__] := U@xx ** (U@xn ** U@yy);
```

```
U[l___, x_^n_, r___] := U[l, Sequence @@ Table[x, {n}], r];
U[l___, 1, r___] := U[l, r];
```

```
𝕆[n_, poly_, specs___] := Module[{vs, us},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
  Total[
   CoefficientRules[Normal@Series[poly, {ħ, 0, n}], vs] /. (p_ → c_) ⧴ c UU @@ (us^p)
  ]
 ]
```

## Testing $\mathfrak{g}_1$

```
LBasis[n_Integer] := LBasis[Range[n]];
LBasis[S_] := DeleteCases[0]@
  Module[{i, j, k, l}, SortBy[(# /. {ϵ → 2, c_ → 2, u_ → 2, w_ → 2, U → Times}) &][
    Union@Flatten[{{U[], ϵ U[]},
       Table[{U@c_i, U@u_i, U@w_i, ϵ U@c_i, ϵ U@u_i, ϵ U@w_i}, {i, S}],
       Table[{U[u_i, w_j], ϵ U[u_i, w_j],
         ϵ U @@ Sort@{c_i, c_j}, ϵ U[u_j, c_i], ϵ U[c_i, w_j]}, {i, S}, {j, S}],
       Table[{ϵ U[u_j, c_i, w_k], ϵ U @@ Sort@{u_i, u_j, w_k}, ϵ U @@ Sort@{u_i, w_j, w_k}},
        {i, S}, {j, S}, {k, S}],
       Table[ϵ U @@ Sort@{u_i, u_j, w_k, w_l}, {i, S}, {j, S}, {k, S}, {l, S}] }]
   ]]
```

```
bas = LBasis[2];
Table[B[x, y] + B[y, x], {x, bas}, {y, bas}] // Flatten // Union
```

{0}

```
bas = LBasis[2]; Timing[
 Table[
    {x, y, z} = xyz;
    Simp[B[B[x, y], z] + B[B[y, z], x] + B[B[z, x], y]],
    {xyz, Subsets[bas, {3}]}
   ] // Flatten // Union
]
```

{29.5156, {0}}

## Testing the Big $\mathfrak{g}_0/\mathfrak{g}_1$ Lemmas

**1.** c Relations (of $\mathfrak{g}_0/\mathfrak{g}_1$).

```
With[{n = 10}, Simp[
   𝕆[n, e^(ħ (γ c_0 + β u_0)), {c_0, u_0} → 0] == 𝕆[n, e^(ħ (γ c_0 + e^(ħ γ) β u_0)), {u_0, c_0} → 0]
  ]]
```

True

```
With[{n = 10}, Simp[
   𝕆[n, e^(ħ (γ c_0 + ħ β w_0)), {w_0, c_0} → 0] == 𝕆[n, e^(ħ (γ c_0 + ħ e^(ħ γ) β w_0)), {c_0, w_0} → 0]
  ]]
```

True

**2.** The Weyl Relations (of $\mathfrak{g}_0$)

```
With[{n = 10}, Simp[
  𝕆[n, ℯ^(ℏ (α w₀+β u₀)), {w₀, u₀} → 0] == 𝕆[n, ℯ^(ℏ (-ℏ α β b₀+α w₀+β u₀)), {u₀, w₀} → 0] /. ϵ → 0
 ]]
```

True

### 3. The Emergence of $\nu$ (in $\mathfrak{g}_0$)

```
With[{n = 10, ν = (1 + ℏ b₀ δ)^-1}, Simp[
  𝕆[n, ℯ^(ℏ δ u₀ w₀) ℯ^(ℏ β u₁), {w₀, u₀, u₁} → 0] == 𝕆[n, ℯ^(ℏ δ u₀ w₀) ℯ^(ℏ ν β u₁), {u₁, w₀, u₀} → 0] /. ϵ → 0
 ]]
```

True

### 4. Reversing a quadratic (in $\mathfrak{g}_0$)

```
With[{n = 10, ν = (1 + ℏ b₀ δ)^-1}, Simp[
  𝕆[n, ℯ^(ℏ δ u₀ w₀), {w₀, u₀} → 0] == 𝕆[n, ν ℯ^(ℏ ν δ u₀ w₀), {u₀, w₀} → 0] /. ϵ → 0
 ]]
```

True

### 5. The main (Λόγος) relation (of $\mathfrak{g}_1$)

```
Λ = 1/2 (ν b₀ (-2 δ^2 - 4 α β δ ν - α^2 β^2 ν^2 + α^2 δ^2 ν^2 w₀^2 + 4 δ^2 ν u₀ w₀ (δ + α β ν + α δ ν w₀) +
        δ^2 ν^2 u₀^2 (β^2 + 4 β δ w₀ + 3 δ^2 w₀^2)) + 2 (2 c₀ (δ + α β ν + α δ ν w₀ + δ ν u₀ (β + δ w₀)) +
        ν^2 (2 δ + α β ν + α δ ν w₀ + δ ν u₀ (β + δ w₀)) (α w₀ + u₀ (β + 2 δ w₀))));
With[{n = 10}, Simp[
  𝕆[n, ℯ^(α w₀+β u₀+δ u₀ w₀) /. {α → ℏ α, β → ℏ β, δ → ℏ δ}, {w₀, u₀} → 0] ==
   𝕆[n, ν (1 + ϵ ν Λ) ℯ^(ν (-b₀ α β+α w₀+β u₀+δ u₀ w₀)) /.
     {ν → (1 + b₀ ℏ δ)^-1, α → ℏ α, β → ℏ β, δ → ℏ δ}, {u₀, c₀, w₀} → 0]
 ]]
```

True

```
Λ1 = Total[CoefficientRules[Λ /. {x_0 :→ x}, {u, c, w}] /.
   (p_ → cc_) :→ Simplify[cc] Times @@ {u, c, w}^p]
```

$$2 c w \alpha \delta \nu + 2 c u \beta \delta \nu + 2 c u w \delta^2 \nu + \frac{1}{2} w^2 \alpha^2 \delta (2 + b \delta) \nu^3 + \frac{1}{2} u^2 \beta^2 \delta (2 + b \delta) \nu^3 + u w^2 \alpha \delta^2 (3 + 2 b \delta) \nu^3 + u^2 w \beta \delta^2 (3 + 2 b \delta) \nu^3 + \frac{1}{2} u^2 w^2 \delta^3 (4 + 3 b \delta) \nu^3 + 2 c (\delta + \alpha \beta \nu) + 2 u w \delta (2 + b \delta) \nu^2 (\delta + \alpha \beta \nu) + w \alpha \nu^2 (2 \delta + \alpha \beta \nu) + u \beta \nu^2 (2 \delta + \alpha \beta \nu) - \frac{1}{2} b \nu (2 \delta^2 + 4 \alpha \beta \delta \nu + \alpha^2 \beta^2 \nu^2)$$

```
Λ1 // TeXForm
```

```
-\frac{1}{2} b \nu  \left(\alpha ^2 \beta ^2 \nu ^2+4 \alpha  \beta  \delta  \nu +2 \delta
   ^2\right)+\frac{1}{2} \beta ^2 \delta  \nu ^3 u^2 (b \delta +2)+\frac{1}{2} \delta ^3 \
   (3 b \delta +4)+\beta  \delta ^2 \nu ^3 u^2 w (2 b \delta +3)+\alpha  \delta ^2 \nu ^3 u
   \delta +3)+2 \delta  \nu ^2 u w (b \delta +2) (\alpha  \beta  \nu +\delta )+\frac{1}{2}
   \delta  \nu ^3 w^2 (b \delta +2)+2 c (\alpha  \beta  \nu +\delta )+2 \beta  c \delta  \n
   \delta ^2 \nu  u w+2 \alpha  c \delta  \nu  w+\beta  \nu ^2 u (\alpha  \beta  \nu +2 \de
   \nu ^2 w (\alpha  \beta  \nu +2 \delta )
```